

# User Guide for **ffr-LFDFT**

Fadjar Fathurrahman

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Installation</b>	<b>1</b>
<b>3</b>	<b>Usage</b>	<b>2</b>
<b>4</b>	<b>Kohn-Sham equation</b>	<b>2</b>
<b>5</b>	<b>Implementation</b>	<b>3</b>
5.1	Main program . . . . .	3
5.2	Description of LF basis set . . . . .	4
5.3	Description of molecular or crystalline structure . . . . .	5
5.4	Pseudopotential . . . . .	5
5.5	Nonlocal pseudopotential . . . . .	6
<b>A</b>	<b>Lagrange basis function</b>	<b>8</b>
A.1	Periodic Lagrange function . . . . .	8
A.2	Cluster Lagrange function . . . . .	9

## 1 Introduction

Welcome to **ffr-LFDFT** documentation.

**ffr-LFDFT** is a poor man's program (or collection of subroutines, as of now) to carry out electronic structure calculations based on density functional theory and Lagrange basis set.

How to compile

How to use

input parameters ...

subroutines ... (implementation)

Add tutorial on how to use **m\_LF3d** module to solve Schrodinger equation in 1d.

In **LF3d** periodic, only gamma-point sampling is used.

## 2 Installation

A manually written **Makefile** is provided. At the topmost part of the **Makefile** you need to specify which **make.inc** file you want to use. You need to decide which compiler to use if there are more than one compiler in you system. In the directory **platform** there are several **make.inc** files. Currently, **ffr-LFDFT** is tested using the following compilers on Linux system:

- GNU Fortran compiler
- G95 Fortran compiler
- Intel Fortran compiler
- PGI Fortran compiler

- Sun (now part of Oracle) Fortran compiler

For typical Linux system, `make.inc.gfortran` is sufficient. You can manually edit the compiler options in the corresponding `make.inc` files.

There following external libraries are required to build `ffr-LFDFT`

- BLAS
- LAPACK
- FFTW3

Typing the command

```
make
```

will build the library `libmain.a` and typing the command

```
make main
```

will build the main executable `ffr_LFDFT.x`.

### 3 Usage

`ffr-LFDFT` main executable, `ffr_LFDFT.x` supports a subset of PWSCF input file.

```
&CONTROL
```

```
/
```

```
&SYSTEM
```

```
/
```

```
&ELECTRONS
```

```
/
```

```
ATOMIC_SPECIES
```

```
...
```

```
ATOMIC_POSITIONS angstrom
```

```
...
```

### 4 Kohn-Sham equation

Central to the density functional theory is the so-called Kohn-Sham equation. This equation can be written as:

$$\left[ -\frac{1}{2}\nabla^2 + V_{\text{KS}}(\mathbf{r}) \right] \psi_i(\mathbf{r}) = \epsilon_i \psi_i(\mathbf{r}) \quad (1)$$

where  $\epsilon_i$  and  $\psi_i(\mathbf{r})$  is known as Kohn-Sham eigenvalues and eigenvectors (orbitals). Quantity  $V_{\text{KS}}$  is called the Kohn-Sham potential, which can be written as sum of several potentials:

$$V_{\text{KS}}(\mathbf{r}) = V_{\text{ion}}(\mathbf{r}) + V_{\text{Ha}}(\mathbf{r}) + V_{\text{xc}}(\mathbf{r}) \quad (2)$$

$V_{\text{ion}}$  denotes attractive potential between ion (or atomic nuclei) with electrons. This potential can be written as:

$$V_{\text{ion}}(\mathbf{r}) = \sum_I^{N_{\text{atoms}}} \frac{Z_I}{|\mathbf{r} - \mathbf{R}_I|} \quad (3)$$

This potential is Coulombic and has singularities at the ionic centers. It is generally difficult to describe this potential fully. It is common to replace the full Coulombic potential with softer potential which is known as pseudopotential. There are various types or flavors of pseudopotentials. In the current implementation, ion-electron potential,  $V_{\text{ion}}$  is treated by pseudopotential. HGH-type pseudopotential is employed due to the the availability of analytic forms both in real and reciprocal space.

$V_{\text{Ha}}$  is the classical Hartree potential. It is defined as

$$V_{\text{Ha}}(\mathbf{r}) = \frac{1}{2} \int \frac{\rho(\mathbf{r})\rho(\mathbf{r}')}{|\mathbf{r} - \mathbf{r}'|} d\mathbf{r}', \quad (4)$$

where  $\rho(\mathbf{r})$  denotes electronic density:

$$\rho(\mathbf{r}) = \sum_i^{N_{\text{occ}}} \psi_i^*(\mathbf{r})\psi_i(\mathbf{r}) \quad (5)$$

Alternatively, Hartree potential can also be obtained via solving Poisson equation:

$$\nabla^2 V_{\text{Ha}}(\mathbf{r}) = -4\pi\rho(\mathbf{r}) \quad (6)$$

The last term in Equation (2) is exchange-correlation potential.

## 5 Implementation

**ffr-LFDFT** is implemented in simple Fortran language. I used global variables heavily, as opposed to using user-defined type to contained them. Currently, only one user-defined type is used in **ffr-LFDFT**, namely `Ps_HGH_Params.T` which is mainly used for convenience. I tried to make the code clear for those who are beginners in implemeting a density-functional calculations (such as myself).

### 5.1 Main program

Currently, the calculation flow of the main program of **ffr-LFDFT** is as follows:

- Getting program argument as input file and reading the input file
- Initializing molecular structure, pseudopotentials, and Lagrange basis functions, including grids
- Setting additional options if necessary based on the input file
- Initializing electronic states variables
- Setting up Hamiltonian: potential and kinetic operators.
- Solving the Kohn-Sham equation via direct minimization or self-consistent field

The appropriate subroutine calls is given below.

```
CALL getarg( 1, filein )
CALL read_input( filein )
CALL setup_from_input()
CALL setup_options()

CALL init_betaNL()
CALL init_states()
CALL init_strfact_shifted()
CALL calc_Ewald_qe()
CALL alloc_hamiltonian()
CALL init_V_ps_loc_G()
CALL init_nabla2_sparse()
CALL init_ilu0_prec()
```

```

CALL gen_guess_rho_gaussian()
ALLOCATE( evecs(Npoints,Nstates), evals(Nstates) )
CALL gen_random_evecs()
CALL gen_gaussian_evecs()

IF( I_KS_SOLVE == 1 ) THEN
  CALL KS_solve_Emin_pcg()
  CALL calc_evals( Nstates, Focc, evecs, evals )
ELSEIF( I_KS_SOLVE == 2 ) THEN
  CALL KS_solve_SCF_v2()
ENDIF

```

## 5.2 Description of LF basis set

Description of LF basis set in 3d is given in module `m_LF3d`. All global variables in this module is given prefix `LF3d`.

```

MODULE m_LF3d
  IMPLICIT NONE
  INTEGER, PARAMETER :: LF3d_PERIODIC = 1
  INTEGER, PARAMETER :: LF3d_CLUSTER = 2
  INTEGER, PARAMETER :: LF3d_SINC = 3
  INTEGER :: LF3d_TYPE
  INTEGER, DIMENSION(3) :: LF3d_NN
  REAL(8), DIMENSION(3) :: LF3d_LL, LF3d_AA, LF3d_BB, LF3d_hh
  INTEGER :: LF3d_Npoints
  REAL(8) :: LF3d_dVol
  REAL(8), ALLOCATABLE :: LF3d_grid_x(:), LF3d_grid_y(:), LF3d_grid_z(:)
  REAL(8), ALLOCATABLE :: LF3d_D1jl_x(:,:), LF3d_D1jl_y(:,:), LF3d_D1jl_z(:,:)
  REAL(8), ALLOCATABLE :: LF3d_D2jl_x(:,:), LF3d_D2jl_y(:,:), LF3d_D2jl_z(:,:)
  REAL(8), ALLOCATABLE :: LF3d_lingrid(:,:)
  INTEGER, ALLOCATABLE :: LF3d_xyz2lin(:,:,:)
  INTEGER, ALLOCATABLE :: LF3d_lin2xyz(:,:)
  REAL(8), ALLOCATABLE :: LF3d_G2(:), LF3d_Gv(:,:)
END MODULE

```

Variables in `m_LF3d` is initialized by calling the subroutine `init_LF3d_XX()`, where `XX` may be one of:

- `p`: periodic LFF
- `c`: cluster LF
- `sinc`: sinc L

```

SUBROUTINE init_LF3d_p( NN, AA, BB )
SUBROUTINE init_LF3d_c( NN, AA, BB )
SUBROUTINE init_LF3d_sinc( NN, hh )

```

In the above subroutines:

- `NN`: an array of 3 integers, specifying sampling points in  $x$ ,  $y$  and  $z$  direction.
- `AA`: an array of 3 floats, specifying left ends of unit cell.
- `BB`: an array of 3 floats, specifying right ends of unit cell.
- `hh`: an array of 3 floats, specifying spacing between adjacent sampling points.

Note that for periodic and cluster LF we have to specify `NN`, `AA`, and `BB` while for sinc LF we have to specify `NN` and `hh`. Note that for periodic LF `NN` must be odd numbers.

Example:

```

NN = (/ 35, 35, 35 /)
AA = (/ 0.d0, 0.d0, 0.d0 /)
BB = (/ 6.d0, 6.d0, 6.d0 /)
CALL init_LF3d_p( NN, AA, BB )

```

### 5.3 Description of molecular or crystalline structure

Description of molecular or crystalline structure is given in module `m_atoms`. Note that unit cell for crystalline structure (currently only orthorhombic structure is possible) is specified by `AA` and `BB` in call to `init_LF3d_p()`

```

MODULE m_atoms
  IMPLICIT NONE
  INTEGER :: Natoms
  INTEGER :: Nspecies
  REAL(8), ALLOCATABLE :: AtomicCoords(:, :)
  INTEGER, ALLOCATABLE :: atm2species(:)
  CHARACTER(5), ALLOCATABLE :: SpeciesSymbols(:)
  REAL(8), ALLOCATABLE :: AtomicValences(:)
  COMPLEX(8), ALLOCATABLE :: StructureFactor(:, :)
END MODULE

```

Currently, variables in module `m_atoms` are initialized by subroutine `init_atoms_xyz()`.

```

SUBROUTINE init_atoms_xyz( fil_xyz )

```

This subroutine takes one argument `fil_xyz` which is the path to XYZ file describing the molecular structure or crystalline structure.

### 5.4 Pseudopotential

Module `m_PsPot`

```

MODULE m_PsPot
  USE m_Ps_HGH, ONLY : Ps_HGH_Params_T
  IMPLICIT NONE
  CHARACTER(128) :: PsPot_Dir = './HGH/'
  CHARACTER(128), ALLOCATABLE :: PsPot_FilePath(:)
  TYPE(Ps_HGH_Params_T), ALLOCATABLE :: Ps_HGH_Params(:)
  INTEGER :: NbetaNL
  REAL(8), ALLOCATABLE :: betaNL(:, :)
  INTEGER, ALLOCATABLE :: prj2beta(:, :, :, :)
  INTEGER :: NprojTotMax
END MODULE

```

We currently support HGH pseudopotential only. The HGH pseudopotential parameter is described by an array of type `Ps_HGH_Params_T` which is defined in `m_Ps_HGH`:

```

TYPE Ps_HGH_Params_T
  CHARACTER(5) :: atom_name
  INTEGER :: zval
  REAL(8) :: rlocal
  REAL(8) :: rc(0:3)
  REAL(8) :: c(1:4)
  REAL(8) :: h(0:3, 1:3, 1:3)
  REAL(8) :: k(0:3, 1:3, 1:3)
  INTEGER :: lmax
  INTEGER :: Nproj_l(0:3) ! number of projectors for each AM
  REAL(8) :: rcut_NL(0:3)
END TYPE

```

## 5.5 Nonlocal pseudopotential

Nonlocal HGH pseudopotential action can be defined as follows:

$$\hat{V}_{NL}\psi = \sum_i^{N_{occ}} \sum_a^{N_{atom}} \sum_{l=0} \sum_{m=-l}^{+l} \sum_{i,j} h_{i,j} \beta_{ialm} \quad (7)$$

Action of nonlocal pseudopotential to wavefunction:

```

SUBROUTINE op_V_ps_NL( Nstates, Vpsi )
  USE m_LF3d, ONLY : Npoints => LF3d_Npoints
  USE m_PsPot, ONLY : NbetaNL, betaNL, prj2beta, Ps => Ps_HGH_Params
  USE m_atoms, ONLY : Natoms, atm2species
  USE m_hamiltonian, ONLY : betaNL_psi
  IMPLICIT NONE
  INTEGER :: Nstates
  REAL(8) :: Vpsi(Npoints,Nstates)
  INTEGER :: ia, isp, ist, ibeta, jbeta, iprj, jprj
  INTEGER :: l, m
  REAL(8) :: hij

  IF( NbetaNL <= 0 ) THEN
    RETURN
  ENDIF

  Vpsi(:, :) = 0.d0

  DO ist = 1, Nstates
    DO ia = 1, Natoms
      isp = atm2species(ia)
      DO l = 0, Ps(isp)%lmax
        DO m = -l, l
          DO iprj = 1, Ps(isp)%Nproj_l(l)
            DO jprj = 1, Ps(isp)%Nproj_l(l)
              ibeta = prj2beta(iprj, ia, l, m)
              jbeta = prj2beta(jprj, ia, l, m)
              hij = Ps(isp)%h(l, iprj, jprj)
              Vpsi(:, ist) = Vpsi(:, ist) + hij*betaNL(:, ibeta)*betaNL_psi(ia, ist, jbeta)
            ENDDO ! jprj
          ENDDO ! iprj
        ENDDO ! m
      ENDDO ! l
    ENDDO
  ENDDO
END SUBROUTINE

```

The array betaNL is defined initialized in subroutine init\_betaNL:

```

SUBROUTINE init_betaNL()

  USE m_LF3d, ONLY : Npoints => LF3d_Npoints, &
    lingrid => LF3d_lingrid, &
    LL => LF3d_LL, &
    dVol => LF3d_dVol
  USE m_PsPot, ONLY : betaNL, NbetaNL, &
    Ps => Ps_HGH_Params
  USE m_atoms, ONLY : atpos => AtomicCoords, Natoms, atm2species
  USE m_Ps_HGH, ONLY : hgh_eval_proj_R
  IMPLICIT NONE

```

```

INTEGER :: ia, isp, l, m, iprj
INTEGER :: Np_beta, ip, ibeta
REAL(8) :: dr_vec(3)
REAL(8) :: dr
REAL(8) :: Ylm_real
REAL(8) :: nrm

ALLOCATE( betaNL(Npoints,NbetaNL) )

! loop structure must be the same as in init_PsPot
ibeta = 0
DO ia = 1,Natoms
  isp = atm2species(ia)
  DO l = 0,Ps(isp)%lmax
    DO iprj = 1,Ps(isp)%Nproj_l(l)
      DO m = -l,l
        ibeta = ibeta + 1
        Np_beta = 0
        DO ip = 1,Npoints
          CALL calc_dr_periodic_1pnt( LL, atpos(:,ia), lingrid(:,ip), dr_vec )
          dr = sqrt( dr_vec(1)**2 + dr_vec(2)**2 + dr_vec(3)**2 )
          IF( dr <= Ps(isp)%rcut_NL(l) ) THEN
            Np_beta = Np_beta + 1
            betaNL(ip,ibeta) = hgh_eval_proj_R( Ps(isp), l, iprj, dr ) * Ylm_real( l, m,
↪ dr_vec )
          ENDIF
        ENDDO
      ENDDO ! iprj
    ENDDO ! m
  ENDDO ! l
ENDDO

END SUBROUTINE

```

and betaNL\_psi is calculated in calc\_betaNL\_psi:

```

SUBROUTINE calc_betaNL_psi( Nstates, psi )

USE m_LF3d, ONLY : Npoints => LF3d_Npoints, &
                  dVol => LF3d_dVol
USE m_PsPot, ONLY : NbetaNL, betaNL
USE m_hamiltonian, ONLY : betaNL_psi
USE m_atoms, ONLY : Natoms
IMPLICIT NONE
INTEGER :: Nstates
REAL(8) :: psi(Npoints,Nstates)
INTEGER :: ist, ibeta, ia
REAL(8) :: ddot

! immediate return if no projectors are available
IF( NbetaNL <= 0 ) THEN
  RETURN
ENDIF

betaNL_psi(:, :, :) = 0.d0

DO ia = 1,Natoms
  DO ist = 1,Nstates
    DO ibeta = 1,NbetaNL

```

```

        betaNL_psi(ia,ist,ibeta) = ddot( Npoints, betaNL(:,ibeta),1, psi(:,ist),1 ) * dVol
    ENDDO
ENDDO
END SUBROUTINE

```

## A Lagrange basis function

### A.1 Periodic Lagrange function

For a given interval  $[0, L]$ , with  $L > 0$ , the grid points  $x_i$  appropriate for periodic Lagrange function are given by:

$$x_i = \frac{L}{2} \frac{2i - 1}{N} \quad (8)$$

with  $i = 1, \dots, N$ . Number of points  $N$  should be an odd number.

The periodic cardinal functions  $L_i^{\text{per}}(x)$ , defined at grid point  $i$  are given by:

$$L_i^{\text{per}}(x) = \frac{1}{\sqrt{NL}} \sum_{n=1}^N \cos\left(\frac{\pi}{L}(2n - N - 1)(x - x_i)\right). \quad (9)$$

The expansion of periodic function in terms of Lagrange functions:

$$f(x) = \sum_{i=1}^N c_i L_i^{\text{per}}(x) \quad (10)$$

with expansion coefficients  $c_i = \sqrt{L/N} f(x_i)$ . When doing variational calculation, the coefficients  $c_i$  are the variational parameters. The actual function values  $f(x_i)$  at grid points  $x_i$  is obtained by  $f(x_i) = \sqrt{N/L} c_i$ . The prefactor is sometimes abbreviated by  $h = L/N$  and is also referred to as scaling factor.

Consider periodic potential in one dimension:

$$V(x + L) = V(x). \quad (11)$$

Floquet-Bloch theorem states that the wave function solution for periodic potentials can be written in the form:

$$\psi_k(x) = e^{ikx} \phi_k(x) \quad (12)$$

where function  $\phi_k(x)$  and its first derivative  $\phi_k'(x)$  have the same periodicity as  $V(x)$  and  $k$  is a constant called the crystal momentum. Substituting this expression to Schrodinger equation we obtain:

$$\left[ -\frac{\hbar^2}{2m} \left( \frac{d^2}{dx^2} + 2ik \frac{d}{dx} - k^2 \right) + V(x) \right] \phi_k(x) = E \phi_k(k). \quad (13)$$

An alternative way of enforcing periodicity of the wave function is to require that:

$$\psi_k(x + L) = e^{ikL} \psi_k(x). \quad (14)$$

This condition follows from:

$$\begin{aligned}
 \psi_k(x + L) &= e^{ik(x+L)} \phi_k(x + L) \\
 &= e^{ik(x+L)} \phi_k(x) \\
 &= e^{ikL} e^{ikx} \phi_k(x) \\
 &= e^{ikL} \psi_k(x)
 \end{aligned}$$



Using periodic cardinal the Schrodinger equation for periodic potential can be written as:

$$\sum_{j=1}^N \left[ -\frac{\hbar^2}{2m} \left( D_{jl}^{(2)} + 2ikD_{jl}^{(1)} - k^2\delta_{jl} \right) + V(j)\delta_{jl} \right] \phi(j) = E\phi(l) \quad (15)$$

with  $l = 1, \dots, N$ .  $D_{jl}^{(1)}$  are matrix elements of the first derivatives:

$$D_{jl}^{(1)} = \begin{cases} 0 & j = l \\ -\frac{2\pi}{L}(-1)^{j-l} \left( 2 \sin \frac{\pi(j-l)}{N} \right)^{-1} & j \neq l \end{cases} \quad (16)$$

and  $D_{jl}^{(2)}$  are matrix elements of the second derivatives,  $N' = (N-1)/2$ :

$$D_{jl}^{(2)} = \begin{cases} -\left(\frac{2\pi}{L}\right)^2 \frac{N'(N'+1)}{3} & j = l \\ -\left(\frac{2\pi}{L}\right)^2 (-1)^{j-l} \frac{\cos(\pi(j-l)/N)}{2 \sin^2[\pi(j-l)/N]} & j \neq l \end{cases} \quad (17)$$

Note that,  $D_{jl}^{(1)}$  is not symmetric, but  $D_{jl}^{(1)} = -D_{lj}^{(1)}$ . Meanwhile, the second derivative matrix  $D_{jl}^{(2)}$  is symmetric, i.e.  $D_{jl}^{(2)} = D_{lj}^{(2)}$ . With the above expressions, first and second derivative of periodic cardinals can be expressed as

$$\frac{d}{dx} L_i^{\text{per}}(x) = \sum_{j=1}^N D_{ji}^{(1)} L_j^{\text{per}}(x) \quad (18)$$

$$\frac{d^2}{dx^2} L_i^{\text{per}}(x) = \sum_{j=1}^N D_{ji}^{(2)} L_j^{\text{per}}(x) \quad (19)$$

The previous approach also can be extended to periodic potential in 3D:

$$V(\mathbf{r}) = V(x, y, z) = V(x + L_x, y + L_y, z + L_z)$$

Using periodic LF, Schrodinger equation can be casted into the following form:

$$\left[ -\frac{\hbar^2}{2m} (\nabla^2 + 2i\mathbf{k} \cdot \nabla - \mathbf{k}^2) + V(\mathbf{r}) \right] \phi_{\mathbf{k}}(\mathbf{r}) = E \phi_{\mathbf{k}}(\mathbf{r}) \quad (20)$$

## A.2 Cluster Lagrange function

For a given interval  $[A, B]$ , with  $B > A$ , the grid points  $x_i$  appropriate for cluster Lagrange function are given by:

$$x_i = A + \frac{B-A}{N+1} i$$

where  $i = 1, \dots, N$ . Number of points  $N$  can be either odd or even number.

The cluster Lagrange functions  $L_i^{\text{clu}}(x)$ , defined at grid point  $i$  are given by:

$$L_i^{\text{clu}}(x) = \frac{2}{\sqrt{(N+1)(B-A)}} \sum_{n=1}^N \sin(k_n(x_i - A)) \sin(k_n(x - A)). \quad (21)$$

where  $k_n = \pi n / (B - A)$ . The expansion of a function  $f(x)$  in terms of cluster Lagrange functions:

$$f(x) = \sum_{i=1}^N c_i L_i^{\text{clu}}(x) \quad (22)$$

with expansion coefficients  $c_i = \sqrt{(B-A)/(N+1)}f(x_i)$ . When doing variational calculation, the coefficients  $c_i$  are the variational parameters. The actual function values  $f(x_i)$  at grid points  $x_i$  is obtained by  $f(x_i) = \sqrt{(N+1)/(B-A)}c_i$ .

Matrix elements  $D_{jl}^{(2)}$  of the second derivatives for cluster Lagrange functions are

$$D_{jl}^{(2)} = \begin{cases} -\frac{1}{2} \left( \frac{\pi}{B-A} \right)^2 \frac{2(N+1)^2 + 1}{3} - \frac{1}{\sin^2 [\pi j / (N+1)]} & j = l \\ -\frac{1}{2} \left( \frac{\pi}{B-A} \right)^2 (-1)^{j-l} \left[ \frac{1}{\sin^2 \left[ \frac{\pi(j-l)}{2(N+1)} \right]} - \frac{1}{\sin^2 \left[ \frac{\pi(j+l)}{2(N+1)} \right]} \right] & j \neq l \end{cases} \quad (23)$$

For free or cluster boundary condition, we don't need  $D_{jl}^{(1)}$ .